

# Excel LAMBDA: Mimic SQL's **INTERSECT** using **BYROW** and **FILTER**

# Consider these two tables of employees

	A	B	C	D	E	F	G
1	Table name: _emp1			Table name: _emp2			
2	<b>first_name</b>		<b>last_name</b>	<b>forename</b>		<b>surname</b>	
3	Guy		Gilbert	Thierry		D'Hers	
4	Kevin		Brown	David		Bradley	
5	Roberto		Tamburello	David		Bradley	
6	Rob		Walters	JoLynn		Dobney	
7	Rob		Walters	Ruth		Ellerbrock	
8	Thierry		D'Hers	Gail		Erickson	
9	David		Bradley	Barry		Johnson	
10	David		Bradley	Jossef		Goldberg	
11	JoLynn		Dobney	Terri		Duffy	
12	Ruth		Ellerbrock	Sidney		Higa	

We want to find the employees that exist in both tables

	A	B	C	D	E	F	G
1	Table name: _emp1			Table name: _emp2			
2	<b>first_name</b>		<b>last_name</b>	<b>forename</b>		<b>surname</b>	
3	Guy		Gilbert	Thierry		D'Hers	
4	Kevin		Brown	David		Bradley	
5	Roberto		Tamburello	David		Bradley	
6	Rob		Walters	JoLynn		Dobney	
7	Rob		Walters	Ruth		Ellerbrock	
8	Thierry		D'Hers	Gail		Erickson	
9	David		Bradley	Barry		Johnson	
10	David		Bradley	Jossef		Goldberg	
11	JoLynn		Dobney	Terri		Duffy	
12	Ruth		Ellerbrock	Sidney		Higa	

We can compare one row from one table with all rows from the other table

	A	B	C	D	E	F	G	H	I
1	Table name: _emp1			Table name: _emp2					
2		<b>first_name</b>	<b>last_name</b>		<b>forename</b>	<b>surname</b>		=B8:C8=_emp2	
3		Guy	Gilbert		Thierry	D'Hers		TRUE	TRUE
4		Kevin	Brown		David	Bradley		FALSE	FALSE
5		Roberto	Tamburello		David	Bradley		FALSE	FALSE
6		Rob	Walters		JoLynn	Dobney		FALSE	FALSE
7		Rob	Walters		Ruth	Ellerbrock		FALSE	FALSE
8		Thierry	D'Hers		Gail	Erickson		FALSE	FALSE
9		David	Bradley		Barry	Johnson		FALSE	FALSE
10		David	Bradley		Jossef	Goldberg		FALSE	FALSE
11		JoLynn	Dobney		Terri	Duffy		FALSE	FALSE
12		Ruth	Ellerbrock		Sidney	Higa		FALSE	FALSE

# Test if all columns in the row are TRUE

**BYROW** iterates through the rows in the array in the first parameter...

	A	B	C	D	E	F	G	H	I	J	K
1		Table name: _emp1			Table name: _emp2						
2		<b>first_name</b>	<b>last_name</b>		<b>forename</b>	<b>surname</b>			<b>=BYROW(B8:C8=_emp2,LAMBDA(r,AND(r)))</b>		
3		Guy	Gilbert		Thierry	D'Hers			TRUE		
4		Kevin	Brown		David	Bradley			FALSE		
5		Roberto	Tamburello		David	Bradley			FALSE		
6		Rob	Walters		JoLynn	Dobney			FALSE		
7		Rob	Walters		Ruth	Ellerbrock			FALSE		
8		Thierry	D'Hers		Gail	Erickson			FALSE		
9		David	Bradley		Barry	Johnson			FALSE		
10		David	Bradley		Jossef	Goldberg			FALSE		
11		JoLynn	Dobney		Terri	Duffy			FALSE		
12		Ruth	Ellerbrock		Sidney	Higa			FALSE		

...and passes each row into the **LAMBDA** in the second parameter

This **LAMBDA** uses **AND** to test if all the columns in the row are TRUE

# Test if any of the rows are TRUE

Passing an array of TRUE/FALSE to **OR** will return TRUE if any value in the array is TRUE, or FALSE if they are all FALSE

	A	B	C	D	E	F	G	H	I	J	K
1		Table name: _emp1			Table name: _emp2						
2		<b>first_name</b>	<b>last_name</b>		<b>forename</b>	<b>surname</b>		=OR(BYROW(B8:C8=_emp2,LAMBDA(r,AND(r))))			
3		Guy	Gilbert		Thierry	D'Hers		TRUE			
4		Kevin	Brown		David	Bradley					
5		Roberto	Tamburello		David	Bradley					
6		Rob	Walters		JoLynn	Dobney					
7		Rob	Walters		Ruth	Ellerbrock					
8		Thierry	D'Hers		Gail	Erickson					
9		David	Bradley		Barry	Johnson					
10		David	Bradley		Jossef	Goldberg					
11		JoLynn	Dobney		Terri	Duffy					
12		Ruth	Ellerbrock		Sidney	Higa					

Apply the **OR** test expression to every row in \_emp1

	A	B	C	D	E	F	G	H	I	J	K	L
1		Table name: _emp1			Table name: _emp2							
2		<b>first_name</b>	<b>last_name</b>		<b>forename</b>	<b>surname</b>		FALSE				
3		Guy	Gilbert		Thierry	D'Hers		FALSE				
4		Kevin	Brown		David	Bradley		FALSE				
5		Roberto	Tamburello		David	Bradley		FALSE				
6		Rob	Walters		JoLynn	Dobney		FALSE				
7		Rob	Walters		Ruth	Ellerbrock		FALSE				
8		Thierry	D'Hers		Gail	Erickson		TRUE				
9		David	Bradley		Barry	Johnson		TRUE				
10		David	Bradley		Jossef	Goldberg		TRUE				
11		JoLynn	Dobney		Terri	Duffy		TRUE				
12		Ruth	Ellerbrock		Sidney	Higa		TRUE				

Advanced Formula Environment

Grid Names Modules

✓ emp1 > H3

```
1 =BYROW(  
2   _emp1,  
3   LAMBDA(emp1row,  
4       OR(  
5           BYROW(  
6               emp1row = _emp2,  
7               LAMBDA(r, AND(r))  
8           )  
9       )  
10  )  
11 )
```

For each row in \_emp1, test if it exists in \_emp2, return TRUE if it does, FALSE otherwise

And pass the resulting array into the 'include' parameter of **FILTER**

	A	B	C	D	E	F	G	H	I
1		Table name: _emp1			Table name: _emp2				
2		<b>first_name</b>	<b>last_name</b>		<b>forename</b>	<b>surname</b>			
3		Guy	Gilbert		Thierry	D'Hers		Thierry	D'Hers
4		Kevin	Brown		David	Bradley		David	Bradley
5		Roberto	Tamburello		David	Bradley		David	Bradley
6		Rob	Walters		JoLynn	Dobney		JoLynn	Dobney
7		Rob	Walters		Ruth	Ellerbrock		Ruth	Ellerbrock
8		Thierry	D'Hers		Gail	Erickson			
9		David	Bradley		Barry	Johnson			
10		David	Bradley		Jossef	Goldberg			
11		JoLynn	Dobney		Terri	Duffy			
12		Ruth	Ellerbrock		Sidney	Higa			

Advanced Formula Environment

Grid Names Modules

✓ emp1 > H3

```
1 =FILTER(  
2   _emp1,  
3   BYROW(  
4     _emp1,  
5     LAMBDA(emp1row,  
6       OR(  
7         BYROW(  
8           emp1row = _emp2,  
9           LAMBDA(r, AND(r))  
10        )  
11      )  
12    )  
13  )  
14 )
```

The first parameter of **FILTER** is the array we want to filter, the second parameter is an array of TRUE/FALSE indicating which rows (or columns) we want to return




# Finally, we can save the formula as a named **LAMBDA** function called **INTERSECT**

Using the function is now very easy

! We can optionally wrap the result in **UNIQUE** to remove duplicate entries

H3             =INTERSECT(\_emp1,\_emp2)

	A	B	C	D	E	F	G	H	I
1		Table name: _emp1			Table name: _emp2				
2		<b>first_name</b>	<b>last_name</b>		<b>forename</b>	<b>surname</b>			
3		Guy	Gilbert		Thierry	D'Hers		Thierry	D'Hers
4		Kevin	Brown		David	Bradley		David	Bradley
5		Roberto	Tamburello		David	Bradley		David	Bradley
6		Rob	Walters		JoLynn	Dobney		JoLynn	Dobney
7		Rob	Walters		Ruth	Ellerbrock		Ruth	Ellerbrock
8		Thierry	D'Hers		Gail	Erickson			
9		David	Bradley		Barry	Johnson			
10		David	Bradley		Jossef	Goldberg			
11		JoLynn	Dobney		Terri	Duffy			
12		Ruth	Ellerbrock		Sidney	Higa			



Advanced Formula Environment

Grid   Names   Modules

Workbook   + New

```
1 INTERSECT = LAMBDA(array1,array2,
2   FILTER(
3     array1,
4     BYROW(
5       array1,
6       LAMBDA(emp1row,
7         OR(
8           BYROW(
9             emp1row = array2,
10            LAMBDA(r, AND(r))
11          )
12        )
13      )
14    )
15  )
16 )
```